#-------------------------------------------------
#
# Project created by QtCreator 2018-10-28T10:44:34
#
#-------------------------------------------------

QT       += core gui

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets
INCLUDEPATH += ../../../tb_interface
DESTDIR = $$OUT_PWD/../../../bin/modules
TARGET = sink_plots
TEMPLATE = app
include (../../../taskbusplatform.pri)

DEFINES += QT_DEPRECATED_WARNINGS
CONFIG += c++17

SOURCES += \
	listen_thread.cpp \
	main.cpp \
	spectrowidget.cpp

HEADERS += \
	listen_thread.h \
	spectrowidget.h


qtHaveModule(charts) {
	message("Qt with charts!");
	QT += charts
	HEADERS += charts/dialogplots.h
	SOURCES += charts/dialogplots.cpp
	FORMS += charts/dialogplots.ui
	INCLUDEPATH += charts
} else:qtHaveModule(graphs){
	message("Qt with graphs!");
	QT += graphs quick quickwidgets
	HEADERS += graphs/dialogplots.h
	SOURCES += graphs/dialogplots.cpp
	FORMS += graphs/dialogplots.ui
	INCLUDEPATH += graphs
}

# Default rules for deployment.
qnx: target.path = /tmp/$${TARGET}/bin
else: unix:!android: target.path = /opt/$${TARGET}/bin
!isEmpty(target.path): INSTALLS += target

DISTFILES += \
	sink_plots.json \
	sink_plots.zh_CN.json

RESOURCES += \
	dialogplots.qrc

QMAKE_POST_LINK += $${QMAKE_COPY} $$PWD/sink_plots.md $$DESTDIR/sink_plots.md
